DFT calculation of the generalized and drazin inverse of a polynomial matrix

N. P. Karampetakis and S. Vologianidis
Department of Mathematics
Aristotle University of Thessaloniki
Thessaloniki 54006, Greece
email : karampet@auth.gr

January 15, 2002

Abstract

A new algorithm is presented for the determination of the a) generalized inverse and b) drazin inverse, of a polynomial matrix. The proposed algorithms are based on the discrete Fourier transform and thus are computationally fast in contrast to other known algorithms. The above algorithms are implemented in the Mathematica programming language and illustrated via examples.

1 Introduction

The definition of the generalized inverse for nonsquare constant matrices firstly has been defined by Penrose [11], while later Decell [2] has been proposed a Leverrier-Faddeev algorithm for its computation. A Leverrier-Faddeev algorithm has also been proposed by Grevile [4] for the computation of the Drazin inverse of square constant matrices. Karampetakis in [8] and later [7], [9] and [12] have proposed new Leverrier algorithms for the determination of the generalized inverse and Drazin inverse of polynomial matrices. These algorithms are good enough if we implement them in symbolic programming languages like Mathematica, Maple e.t.c.. However their main disadvantage is the same with all the known Leverrier algorithms : are not stable if they are implemented in other high level programming languages such as C++, Fortran e.t.c.

During the past two decades there has been extensive use of Discrete Fourier Transform (DFT) - based algorithms, due to their computational speed and accuracy. Some remarkable examples, but not the only, of the use of DFT in linear algebra problems are the calculation of the determinantal polynomial by [10], the computation of the transfer function of generalized n-dimensional systems by [1] and the solutions of polynomial matrix diophantine equations by [6].

The reason for the interest in these two specific inverses are due to their applications in inverse systems, solution of AutoRegressive Moving Average representations [5] and solution of diophantine equations which gives rise to numerous applications (see for example [8] and its references).

The main purpose of this work is to present a DFT-algorithm for the evaluation of the generalized inverse and the Drazin inverse of a polynomial matrix. More specifically in section 2 we introduce the 2-dimensional discrete Fourier transform, while later in section 3 and 4 we propose two new DFT algorithms for the evaluation of the generalized and Drazin inverse respectively of a polynomial matrix. Finally in section 5 we produce a code for the implementation of these algorithms. The whole theory is illustrated via examples and tested in comparison to the methods presented in [9] in section 6.

2 The discrete Fourier transform

Consider the finite sequence $X(k)$ and $\tilde{X}(r)$ $k, r = 0, 1, ..., M$. In order for the sequence $X(k)$ and $\tilde{X}(r)$ to constitute an DFT pair the following relations should hold [3] :

$$\tilde{X}(r) = \sum_{k=0}^{M} X(k) W^{-kr} \tag{1}$$

$$X(k) = \frac{1}{M+1} \sum_{r=0}^{M} \tilde{X}(r) W^{kr} \tag{2}$$

where

$$W = e^{\frac{2\pi j}{M+1}} \tag{3}$$

$X, \tilde{X}$ are discrete argument matrix-valued functions, with dimensions $p \times m$.

Consider now the finite sequence $X(k_1, k_2)$ and $\tilde{X}(r_1, r_2)$, $k_i, r_i = 0, 1, ..., M_i, i = 1, 2$. In order for the sequence $X(k_1, k_2)$ and $\tilde{X}(r_1, r_2)$ to constitute an DFT pair the following relations should hold [3] :

$$\tilde{X}(r_1, r_2) = \sum_{k_1=0}^{M_1} \sum_{k_2=0}^{M_2} X(k_1, k_2) W_1^{-k_1 r_1} W_2^{-k_2 r_2} \tag{4}$$

$$X(k_1, k_2) = \frac{1}{R} \sum_{r_1=0}^{M_1} \sum_{r_2=0}^{M_2} \tilde{X}(r_1, r_2) W_1^{k_1 r_1} W_2^{k_2 r_2} \tag{5}$$

where

$$\begin{aligned} R &= (M_1 + 1) \times (M_2 + 1) \\ W_i &= e^{\frac{2\pi j}{M_i + 1}}, i = 1, 2 \end{aligned} \tag{6}$$

$X, \tilde{X}$ are discrete argument matrix-valued functions, with dimensions $p \times m$.

3 Generalized inverse of a polynomial matrix

Consider the polynomial matrix

$$A(s) = A_q s^q + \cdots = A_1 s + A_0 \in R[s]^{p\times m} \tag{7}$$

with $A_i \in R^{p\times m}, i \in q$, and p not necessarily equal to m.

Definition 1 *[11] For every matrix $A \in R^{p\times m}$, a unique matrix $A^+ \in R^{m\times p}$, which is called generalized inverse, exists satisfying*

(i) $AA^+A = A$

(ii) $A^+AA^+ = A^+$

(iii) $(AA^+)^T = AA^+$

(iv) $(A^+A)^T = A^+A$

where A^T denotes the trasnpose of A. In the special case that the matrix A is square nonsingular matrix, the generalized inverse of A is simply its inverse i.e. $A^+ = A^{-1}$.

In an analogous way we define the generalized inverse $A(s)^+ \in R(s)^{m\times p}$ of the polynomial matrix $A(s) \in R[s]^{p\times m}$ defined in (7) as the matrix which satisfies the properties (i)-(iv) of Definition 1. [8] proposed the following Theorem for the computation of the generalized inverse of a polynomial matrix (7).

Theorem 2 *[8] Let $A(s) \in R[s]^{p\times m}$ as in (7) and*

$$\begin{aligned} a(s,z) &= \det\left[zI_p - A(s)A(s)^T\right] \\ &= a_0(s)z^p + a_1(s)z^{p-1} + \cdots + a_{p-1}(s)z + a_p(s) \end{aligned} \tag{8}$$

$a_0(s) = 1$, be the characteristic polynomial of $A(s)A(s)^T$. Let $a_p(s) \equiv 0, ..., a_{k+1}(s) \equiv 0$ while $a_k(s) \neq 0$ and $\Lambda := \{s_i \in R : a_k(s_i) = 0\}$. Then the generalized inverse $A(s)^+$ of $A(s)$ for $s \in R - \Lambda$ is given by

$$\begin{aligned} A(s)^+ &= -a_k(s)A(s)^T\left[\left(A(s)A(s)^T\right)^{k-1} + a_1(s)\left(A(s)A(s)^T\right)^{k-2} + \cdots + a_{k-1}(s)I_p\right] \\ &= -a_k(s)A(s)^T B_{k-1}(s) \end{aligned} \tag{9}$$

If $k = 0$ is the largest integer such that $a_k(s) \neq 0$, then $A(s)^+ = 0$. For those $s_i \in \Lambda$ find the largest integer $k_i < k$ such that $a_{k_i}(s_i) \neq 0$ and then the generalized inverse $A(s_i)^+$ of $A(s_i)$ is given by

$$\begin{aligned} A(s_i)^+ &= -a_{k_i}(s_i)A(s_i)^T\left[\left(A(s_i)A(s_i)^T\right)^{k-1} + a_1(s_i)\left(A(s_i)A(s_i)^T\right)^{k-2} + \cdots + a_{k_i-1}(s_i)I_p\right] \\ &= -a_{k_i}(s_i)A(s_i)^T B_{k_i-1}(s_i) \end{aligned}$$

Although the above algorithm is good enough for a symbolic programming language, it is not the appropriate for a high level programming language. Therefore in the sequel we propose a 4 step algorithm for the computation of the generalized inverse of $A(s)$ through DFT transforms.

Algorithm 3 *(Evaluation of the generalized inverse of $A(s)$)*

Step 1. (Evaluation of the polynomial $a(s,z)$)

It is easily seen from (8), that the greatest power n_1 of s in $a(s,z)$ is equal to the greatest power among the powers of $a_i(s), i = 1,2,..,p$. Note however, [8] that the greatest power of $a_k(s)$ is $2kq$ i.e. $n_1 = \max\{2kq, k = 1,2,...,p\} = 2pq$. The greatest power n_2 of z in $a(s,z)$ is p i.e. $n_2 = p$. Thus the polynomial $a(s,z)$ can be written as

$$a(s,z) = \sum_{l_1=0}^{n_1} \sum_{l_2=0}^{n_2} a_{l_1,l_2} s^{l_1} z^{l_2} \tag{10}$$

The polynomial $a(s,z)$ can be numerically computed using the following $R = (2pq+1) \times (p+1)$ points

$$\begin{gathered} u_i(r_j) = W_i^{-r_j}, i = 1,2 \\ i = 1,2 \text{ and } r_j = 0,1,...,M_i \end{gathered} \tag{11}$$

where

$$\begin{gathered} W_i = e^{\frac{2\pi j}{M_i+1}} \\ i = 1,2 \;\; ; \;\; M_1 = 2pq \;\; ; \;\; M_2 = p \end{gathered}$$

To evaluate the coefficients a_{r_1,r_2} define

$$\tilde{a}_{r_1,r_2} = \det[u_2(r_2) I_p - A(u_1(r_1)) A(u_1(r_1))^T] \tag{12}$$

From equations (10), (11) and (12) it follows that

$$\tilde{a}_{r_1,r_2} = \sum_{l_1=0}^{n_1} \sum_{l_2=0}^{n_2} a_{l_1,l_2} W_1^{-r_1 l_1} W_2^{-r_2 l_2} \tag{13}$$

Using equations (13) and (5) it is obvious that $[\tilde{a}_{r_1,r_2}]$ and $[a_{l_1,l_2}]$ form a DFT pair. Therefore the coefficients $[a_{l_1,l_2}]$ can be computed using the inverse 2-D DFT as follows

$$a_{l_1,l_2} = \frac{1}{R} \sum_{r_1=0}^{n_1} \sum_{r_2=0}^{n_2} \tilde{a}_{r_1,r_2} W_1^{r_1 l_1} W_2^{r_2 l_2}$$

where $l_1 = 0,1,...,2pq$ and $l_2 = 0,1,..,p$.

Step 2. (Evaluate $a_k(s)$)

Find $k : a_{k+1}(s) = a_{k+2}(s) = \cdots = a_p(s) = 0$ and $a_k(s) \neq 0$
or $a_{l_1,0} = a_{l_1,1} = \cdots = a_{l_1,k+1} = 0 \; \forall l_1$ and $a_{l_1,k} \neq 0$ for some k.

Step 3. (Evaluate $A(s)^T B_{k-1}(s)$)

It is easily seen that the greatest power n of s in

$$B(s) = A(s)^T B_{k-1}(s) = A(s)^T \left[\left(A(s)A(s)^T\right)^{k-1} + a_1(s)\left(A(s)A(s)^T\right)^{k-2} + \cdots + a_{k-1}(s)I_p \right]$$

is $n = \max\{2(k-1)q + q, k = 1, 2, ..., p\} = (2p-1)q$. Thus the polynomial matrix $B(s)$ can be written as

$$B(s) = \sum_{l=0}^{n} B_l s^l \tag{14}$$

The polynomial $B(s)$ can be numerically computed using the following $R = (2p-1)q + 1$ points

$$u(r) = W^{-r} \tag{15}$$

where

$$W = e^{\frac{2\pi j}{(2p-1)q+1}}$$

To evaluate the coefficients B_l define

$$\tilde{B}_r = B(u(r)) \tag{16}$$

From equations (14), (15) and (16) it follows that

$$\tilde{B}_r = \sum_{l=0}^{n} B_l W^{-lr} \tag{17}$$

Using equations (17) and (2) it is obvious that $[\tilde{B}_i]$ and $[B_l]$ form a DFT pair. Therefore the coefficients $[B_l]$ can be computed using the inverse DFT as follows

$$B_l = \frac{1}{R} \sum_{r=0}^{n} \tilde{B}_r W^{lr}$$

where $l = 0, 1, ..., (2p-1)q$.

Step 4. (Evaluation of the generalized inverse)

$$A(s)^+ = \frac{B(s)}{-a_k(s)}$$

4 Drazin inverse of a polynomial matrix

Using the same approach with the previous section, we define the Drazin inverse of a polynomial matrix and find a DFT algorithm for its computation.

Definition 4 *For every matrix $A \in R^{m\times m}$, a unique matrix $A^D \in R^{m\times m}$, which is called Drazin inverse, exists satisfying*

(i) $A^{k+1}A^D = A^k$ *for* $k = ind(A) = \min(k \in N : rank\left(A^k\right) = rank\left(A^{k+1}\right))$

(ii) $A^D A A^D = A^D$

(iii) $AA^D = A^D A$

In the special case that the matrix A is square nonsingular matrix, the Drazin inverse of A is simply its inverse i.e. $A^D = A^{-1}$.

In an analogous way we define the Drazin inverse $A(s)^D \in R(s)^{m\times m}$ of the polynomial matrix $A(s) \in R[s]^{m\times m}$ defined in (7) (with $p = m$) as the matrix which satisfies the properties (i)-(iii) of Definition 4. [12] proposed the following algorithm for the computation of the Drazin inverse of a polynomial matrix (7).

Theorem 5 *[12] Consider a nonregular one-variable rational matrix $A(s)$. Assume that*

$$a(z,s) = \det\left[zI_m - A(s)\right] \quad = a_0(s)z^m + a_1(s)z^{m-1} + \cdots a_{m-1}(s)z + a_m(s), \qquad a_0(s) \equiv 1, \quad z \in \mathbb{C}$$

is the characteristic polynomial of $A(s)$. Also, consider the following sequence of $m \times m$ polynomial matrices

$$B_j(s) = a_0(s)A(s)^j + a_1(s)A(s)^{j-1} + \cdots a_{j-1}(s)A(s) + a_j(s)I_m, \qquad a_0(s) = 1, \quad j = 0, \ldots, m$$

Let

$$a_m(s) \equiv 0, \ldots, a_{t+1}(s) \equiv 0, \quad a_t(s) \neq 0.$$

Define the following set:

$$\Lambda = \{s_i \in \mathbb{C}\text{: } a_t(s_i) = 0\}$$

Also, assume

$$B_m(s) \equiv \nvdash, \ldots, B_r(s) = 0, B_{r-1}(s) \neq 0$$

and $k = r - t$. In the case $s \in C\backslash\Lambda$ and $k > 0$, the Drazin inverse of $A(s)$ is given by

$$\begin{aligned} A(s)^D &= (-1)^{k+1}a_t(s)^{-k-1}A(s)^k B_{t-1}(s)^{k+1} = \\ &= (-1)^{k+1}a_t(s)^{-k-1}A(s)^k \left[a_0(s)A(s)^{t-1} + \cdots + a_{t-2}(s)A(s) + a_{t-1}(s)I_m\right]^{k+1} \end{aligned}$$

In the case $s \in C\backslash\Lambda$ and $k = 0$, we get $A(s)^D = O$.

For $s_i \in \Lambda$ denote by t_i the largest integer satisfying $a_{t_i}(s_i) \neq 0$, and by r_i the smallest integer satisfying $B_{r_i}(s_i) \equiv O$. Then the Drazin inverse of $A(s_i)$ is equal to

$$\begin{aligned} A(s_i)^D &= (-1)^{k_i+1} a_{t_i}(s_i)^{-k_i-1} A(s_i)^{k_i} B_{t_i-1}(s_i)^{k_i+1} = \\ &= (-1)^{k_i+1} a_{t_i}(s_i)^{-k_i-1} A(s_i)^{k_i} \left[a_0(s_i)A(s_i)^{t_i-1} + \cdots + a_{t_i-2}(s_i)A(s_i) + a_{t_i-1}(s_i)I_m\right]^{k_i+1} \end{aligned}$$

where $k_i = r_i - t_i$. ∎

It is known that a q-th order polynomial is zero iff its value at q+1 points is zero. The same holds for polynomial matrices as we can easily in the following Lemma.

Lemma 6 *A polynomial matrix $B(s) = B_0 + B_1 s + \cdots + B_q s^q \in R[s]^{m\times m}$ is the zero polynomial matrix iff its value at $q+1$ distinct points is the zero matrix.*

Proof. Consider that the value of the matrix $B(s)$ at the $q+1$ distinct points $s_i, i = 0, 1, ..., q+1$ is zero. Then we shall have that

$$\begin{bmatrix} B(s_0) & B(s_1) & \cdots & B(s_q) \end{bmatrix} =$$

$$= \begin{bmatrix} B_0 & B_1 & \cdots & B_q \end{bmatrix} \underbrace{\begin{bmatrix} I_m & I_m & \cdots & I_m \\ s_0 I_m & s_1 I_m & \cdots & s_q I_m \\ \vdots & \vdots & \ddots & \vdots \\ s_0^q I_m & s_1^q I_m & \cdots & s_q^q I_m \end{bmatrix}}_{R} = 0_{[m]\times[(q+1)m]}$$

However since the $q+1$ points are distinct the Vandermode matrix R has nonzero determinant and therefore the above system of equations has the unique solution $B_i = 0, i = 0, 1, ..., q.$ ∎

In what follows, we propose a 6-step algorithm for the evaluation of the Drazin inverse of a polynomial matrix $A(s)$.

Algorithm 7 *(Evaluation of the Drazin inverse of a polynomial matrix $A(s)$)*

Step 1. (Evaluation of the polynomial $a(s, z)$)

It is easily seen that the greatest power n_1 of s in $a(s, z)$ is equal to the greatest power among the powers of $a_i(s), i = 1, 2, .., m$. Note [9] that the greatest power of $a_k(s)$ is $2kq$ i.e. $n_1 = \max\{2kq, k = 1, 2, ..., m\} = 2mq$. The greatest power n_2 of z in $a(s, z)$ is m i.e. $n_2 = m$. Thus the polynomial $a(s, z)$ can be written as

$$a(s,z) = \sum_{l_1=0}^{n_1} \sum_{l_2=0}^{n_2} a_{l_1,l_2} s^{l_1} z^{l_2} \tag{18}$$

The polynomial $a(s, z)$ can be numerically computed using the following $R = (2mq + 1) \times (m + 1)$ points

$$u_i(r_i) = W_i^{-r_i}, i = 1, 2 \tag{19}$$

where

$$W_i = e^{\frac{2\pi j}{M_i+1}}$$
$$i = 1, 2 \ ; \ M_1 = 2mq \ ; \ M_2 = m$$

To evaluate the coefficients a_{l_1,l_2} define

$$\tilde{a}_{r_1,r_2} = \det[u_2(r_2)I_m - A(u_1(r_1))] \tag{20}$$

From equations (18), (19) and (20) it follows that

$$\tilde{a}_{r_1,r_2} = \sum_{l_1=0}^{n_1} \sum_{l_2=0}^{n_2} a_{l_1,l_2} W_1^{-r_1 l_1} W_2^{-r_2 l_2} \tag{21}$$

Using equations (21) and (5) it is obvious that $[\tilde{a}_{r_1,r_2}]$ and $[a_{l_1,l_2}]$ form a DFT pair. Therefore the coefficients $[a_{l_1,l_2}]$ can be computed using the inverse 2-D DFT as follows

$$a_{l_1,l_2} = \frac{1}{R} \sum_{r_1=0}^{n_1} \sum_{r_2=0}^{n_2} \tilde{a}_{r_1,r_2} W_1^{r_1 l_1} W_2^{r_2 l_2}$$

where $l_1 = 0, 1, ..., 2mq$ and $l_2 = 0, 1, .., m$.

Step 2. (Evaluate $a_t(s)$)
Find $t : a_{t+1}(s) = a_{t+2}(s) = \cdots = a_m(s) = 0$ and $a_t(s) \neq 0$
or $a_{r_1,0} = a_{r_1,1} = \cdots = a_{r_1,t+1} = 0 \ \forall r_1$ and $a_{r_1,t} \neq 0$ for some t.

Step 3. (Evaluate $r \geq t : B_m(s) \equiv 0$,...,$B_r(s) \equiv 0, B_{r-1}(s) \neq 0$)
Now using lemma 6, we can easily find an algorithm for the determination of r. More specifically consider the polynomial matrix

$$B_j(s) = A(s)^j + a_1(s)A(s)^{j-1} + \cdots a_{j-1}(s)A(s) + a_j(s)I_m$$

The greatest power n of s in $B_j(s)$ is $n_j = 2jq$. In order now to determine the value of $r \geq t$ which satisfy the property : $B_m(s) \equiv \nvdash$,...,$B_r(s) \equiv 0, B_{r-1}(s) \neq 0$ we use the following short algorithm :

$j = m$
Determine the value of $B_j(s)$ at the following $n_j + 1$ points (or any other $n_j + 1$ distinct points)

$$u(r) = W^{-r}, W = e^{\frac{2\pi j}{n_j+1}}$$

Do WHILE $(B_j(s) = 0 \ \forall u(r))$

$j = j - 1$
Determine the value of $B_j(s)$ at the following $n_j + 1$ points

$$u(r) = W^{-r}, W = e^{\frac{2\pi j}{n_j+1}}$$

END DO
$r = j$

The scepticism of the above short algorithm is that the polynomial matrix $B_j(s)$ of degree n_j coincides with the zero matrix if its value at $n_j + 1$ points is equal to zero (Lemma 6).

Step 4. (Evaluation of $A(s)^k B_{t-1}(s)^{k+1}$)
Let $k = r - t$. Then the greatest power n of s in

$$\begin{aligned} B(s) &= A(s)^k B_{t-1}(s)^{k+1} = \\ &= A(s)^k \left[a_0(s)A(s)^{t-1} + \cdots + a_{t-2}(s)A(s) + a_{t-1}(s)I_m\right]^{k+1} \end{aligned}$$

is $n = 2(t-1)q(k+1)+qk$. Thus the polynomial matrix $B(s)$ can be written as

$$B(s) = \sum_{l=0}^{n} B_l s^l \tag{22}$$

The polynomial matrix $B(s)$ can be numerically computed using the following $R = n + 1$ points

$$u(r) = W^{-r} \tag{23}$$

where

$$W = e^{\frac{2\pi j}{n+1}}$$

To evaluate the coefficients B_l define

$$\tilde{B}_r = B(u(r)) \tag{24}$$

From equations (22), (23) and (24) it follows that

$$\tilde{B}_r = \sum_{l=0}^{n} B_l W^{-lr} \tag{25}$$

Using equations (25) and (2) it is obvious that $[\tilde{B}_i]$ and $[B_l]$ form a DFT pair. Therefore the coefficients $[B_l]$ can be computed using the inverse DFT as follows

$$B_l = \frac{1}{R}\sum_{r=0}^{n} \tilde{B}_l W^{lr}$$

where $l = 0, 1, ..., n$.

Step 5. (Evaluation of $a_t(s)^{k+1}$)
The greatest power n of s in

$$a(s) = a_t(s)^{k+1}$$

is $n = 2tq(k+1)$. Thus the polynomial $a(s)$ can be written as

$$a(s) = \sum_{l=0}^{n} a_l s^l \tag{26}$$

The polynomial $a(s)$ can be numerically computed using the following $R = n+1$ points

$$u(r) = W^{-r} \tag{27}$$

where

$$W = e^{\frac{2\pi j}{n+1}}$$

To evaluate the coefficients a_l define

$$\tilde{a}_r = a(u(r)) \tag{28}$$

From equations (26), (27) and (28) it follows that

$$\tilde{a}_r = \sum_{l=0}^{n} a_l W^{-lr} \tag{29}$$

Using equations (29) and (2) it is obvious that $[\tilde{a}_l]$ and $[a_l]$ form a DFT pair. Therefore the coefficients $[a_l]$ can be computed using the inverse DFT as follows

$$a_l = \frac{1}{R}\sum_{r=0}^{n} \tilde{a}_r W^{lr}$$

where $l = 0, 1, ..., n$.

Step 6. (Evaluation of the Drazin inverse)

$$A(s)^D = \frac{B(s)}{a(s)}$$

5 Implementation

In this section we will briefly describe the implementation of the Algorithm 3 and Algorithm 7 in the package Mathematica. The main procedures we use are the following :

Procedure : GeneralizedInverse[A_]
Parameters : "A_" - polynomial matrix $A(s)$
Use : Compute the generalized inverse of $A(s)$ using the Algorithm 3.

Procedure : DrazinInverse[A_]
Parameters : "A_" - polynomial matrix $A(s)$
Use : Compute the Drazin inverse of $A(s)$ using the Algorithm 7.

The code of the above procedures as well as the code of other necessary procedures are given in Appendix 1.

Example 8 *Consider the polynomial matrix*

$$A(s) = \begin{bmatrix} 1 & s & 0 \\ 0 & 1 & s \end{bmatrix} \in R[s]^{2\times 3}$$

The normal inverse of $A(s)$ is clearly not defined; however, the generalized inverse of $A(s)$ can be computed via the implementation of the procedure GeneralizedInverse[A_]. This is illustrated below

```
In[20]:= A = {{1, s, 0}, {0, 1, s}}
```

$$\text{Out[20]=} \begin{pmatrix} 1 & s & 0 \\ 0 & 1 & s \end{pmatrix}$$

```
In[21]:= GeneralizedInverse[A]
```

$$\text{Out[21]=} \begin{pmatrix} \frac{s^2+1}{s^4+s^2+1} & -\frac{s}{s^4+s^2+1} \\ \frac{s^3}{s^4+s^2+1} & \frac{1}{s^4+s^2+1} \\ -\frac{s^2}{s^4+s^2+1} & \frac{s^3+s}{s^4+s^2+1} \end{pmatrix}$$

Example 9 *Consider the polynomial matrix*

$$A(s) = \begin{bmatrix} 1+s & s & 1+s \\ s & -1+s & s \\ 1+s & s & 1+s \end{bmatrix} \in R[s]^{2\times 3}$$

The normal inverse of $A(s)$ is clearly not defined, since $A(s)$ is singular; however, the Drazin inverse of $A(s)$ can be computed via the implementation of the procedure $DrazinInverse[A\_]$. This is illustrated below

```
In[22]:= a = {{s+1, s, s+1}, {s, -1+s, s}, {s+1, s, s+1}}
```

$$\text{Out[22]=} \begin{pmatrix} s+1 & s & s+1 \\ s & s-1 & s \\ s+1 & s & s+1 \end{pmatrix}$$

```
In[23]:= DrazinInverse[a]
```

$$\text{Out[23]=} \begin{pmatrix} \frac{1-s}{4} & \frac{s}{2} & \frac{1-s}{4} \\ \frac{s}{2} & \frac{1}{4}(-4s-4) & \frac{s}{2} \\ \frac{1-s}{4} & \frac{s}{2} & \frac{1-s}{4} \end{pmatrix}$$

The efficiency of the algorithms have been evaluated using the Mathematica function "Timing" which returns the CPU time consumed in seconds. All the tests run on a COMPAQ Presario with CPU a Pentium III at 700MHz and 128Mb of memory, using Windows 2000 Professional and Mathematica 4.1. The discrete Fourier transform based algorithm for the generalized inverse (Drazin inverse) in the following will be denoted by DFTGI (DFTDI). Additionally, our implementation of Generalized (Drazin) inverse is compared with the implementation of [12] ([8]).

5.1 Generalized Inverse

For the test we used random matrices up to dimension 5 and degree 5 and in the following the CPU times from the DFTDI and [8] are shown in the third and fourth columns.

Rows (p)	**Degree** (d)	**DFTGI**	**Karampetakis** [8]
2	0	0.055	0.045
3	0	0.107	0.103
4	0	0.050	0.025
5	0	0.110	0.070
2	1	0.105	0.082
3	1	0.120	0.080
4	1	0.131	0.100
5	1	0.040	0.010
2	2	0.175	0.111
3	2	0.077	0.043
4	2	0.310	0.291
5	2	0.511	0.641
2	3	0.133	0.085
3	3	0.340	0.361
4	3	0.130	0.075
5	3	0.301	0.220
2	4	0.328	0.296
3	4	0.560	0.618
4	4	0.606	0.961
5	4	0.151	0.070
2	5	0.826	1.001
3	5	0.281	0.270
4	5	1.382	2.724
5	5	2.304	7.120

After a basic statistical analysis the algorithm CPU time proved to be linear with $b = (p\log(d+1))^3$ where p the number of rows and d the degree of the matrix, with adjusted $R^2 = 0.991$. The coefficient of b was 0.0342. The following graph shows the dependence of the CPU time (in both algorithms), the number of rows and the degree of the involved polynomial matrix. In order to find the functions which best fit the data presented in the previous table plot, we used the program TableCurve3D. Thus we found two plots one for the DFTGI algorithm and one for the [8]-algorithm. It is easily seen that for small values of the degree and the size of the polynomial matrix, the algorithm presented on [8] is better, while for bigger values of the degree and the size the DFTGI algorithm gives better results i.e. the plot of the CPU time in DFTGI is under the plot of the CPU-time in [8]-algorithm.



Diagram 1. Graph of the CPU time=F(rows,degree)
in both algorithms DFTGI and [8].

5.2 Drazin Inverse

For the test we used random matrices up to dimension 5 and degree 4 and in the following the CPU times from the DFTDI and [12] are shown in the third and fourth columns.

Dimension (p)	**Degree (d)**	**DFTDI**	**Stanimirovic & Karampetakis [12]**
2	0	0.010	0.000
2	1	0.050	0.040
2	2	0.090	0.070
2	3	0.120	0.080
2	4	0.181	0.120
3	0	0.020	0.000
3	1	0.090	0.140
3	2	0.200	0.231
3	3	0.310	0.301
3	4	0.430	0.391
4	0	0.030	0.010
4	1	0.180	0.371
4	2	0.390	0.851
4	3	0.651	1.002
4	4	0.951	1.052
5	0	0.040	0.010
5	1	0.330	0.912
5	2	0.711	1.392
5	3	1.191	2.324
5	4	1.802	2.724

The efficiency of the algorithm is obvious as the dimensions of the matrices get bigger. After a basic statistical analysis the algorithm CPU time proved to be linear with $a = (p \log(d+1))^3$ where p the number of rows and d the degree of the matrix with adjusted $R^2 = 0.996$. The coefficient of a was 0.03436. Similarly, with the previous method for the generalized inverse, we construct the following graph which shows the dependence of the CPU time (in both algorithms), the number of rows and the degree of the involved polynomial matrix. The same conclusions with the DFTGI method are also hold here.



Diagram 2. Graph of the CPU time=F(rows,degree)
in both algorithms DFTDI and [12]

6 Conclusions

In this paper we have presented two new algorithms for the computation of the generalized inverse and Drazin inverse of a polynomial matrix. The proposed algorithms tested and shown to be more efficient from the known ones of [8] and [12] in the case where the degree and the size of the polynomial matrix gets bigger. However, this was expected since the algorithms are based on the known discrete fourier transform. The proposed algorithms can be easily extened to the multivariable polynomial matrices. The clear benefit of computing such a genarlized (Drazin) inverse is that it enables a wider set of such problems to be solved [7].

References

[1] G. E. Antoniou, 2001, Transfer function computation for generalized n-dimensional systems, Journal of the Franklin Institute, 338, 83-90.

[2] H.P. Decell, An application of the Cayley-Hamilton theorem to generalized matrix inversion, SIAM Review, 7, No 4, 1965, 526–528.

[3] D.E. Dudgeon, R.M. Mersereau, 1984, Multidimensional Digital Signal Processing, Prentice-Hall, Englewood Cliffs, N.J..

[4] T.N.E. Grevile, The Souriau-Frame algorithm and the Drazin pseudoinverse, Linear Algebra Appl., 6, 1973, 205–208.

[5] T. Guyer, O. Kiymaz, G. Bilgici and S. Mirasyedioglu, 2001, A new method for computing the solutions of differential equation systems using generalized inverse via Maple, Applied Mathematics and Computation, 121, 291-299.

[6] M. Hromcik and M. Sebek, 2001, Fast Fourier tranaform and linear polynomial matrix equations, Proceedings of the 1rst IFAC Workshop on Systems Structure and Control, Prague, Czech Republic.

[7] J. Jones, N.P. Karampetakis and A.C. Pugh, The computation and application of the generalized inverse via Maple, J. Symbolic Computation, 25, 1998, 99–124 .

[8] N.P. Karampetakis, 1997, Computation of the generalized inverse of a polynomial matrix and applications., Linear Algebra and its Applications, 252, 35-60.

[9] N. Karampetakis and P. Stanimirovic, 2001, On the computation of the Drazin inverse of a polynomial matrix, 1rst IFAC Symposium on Systems Structure and Control, Prague, Czech Republic.

[10] L. E. Paccagnella and G. L. Pierobon, 1976, FFT calculation of a determinantal polynomial, IEEE Trans. on Automatic Control, pp.401-402.

[11] R. Penrose, 1955, A generalized inverse for matrices, Proc. Cambridge Philos. Soc., 51, 406-413.

[12] P. Stanimirovic and N.P. Karampetakis, 2000, Symbolic implementation of Leverrier-Faddeev algorithm and applications., 8th IEEE Medit. Conference on Control and Automation, Patra, Greece.

Appendix 1. Mathematica Code

```
FourierPoints[x_Integer, ri_Integer, p_Integer, q_Integer] :=
 Module[{M, lc},
  M = {2 * p * q, p, (2 * p - 1) * q, 2 * (p - 1) * q};
  lc = Chop[N[Exp[(-2 * ri * Pi * I) / (M[[x]] + 1)]], 10^(-10)]
 ]

CalculateDeterminantCoefficient[A_, i1_Integer,
  i2_Integer, p_Integer, m_Integer, q_Integer, opts___?OptionQ] :=
 Module[{deflt, method, res},
  deflt = {Method → GeneralizedInverse};
  {method} = {Method} /. Flatten[{opts}] /. deflt;
  Switch[method,
   GeneralizedInverse,
       res = Chop[Det[FourierPoints[2, i2, p, q] * IdentityMatrix[p] -
         (A /. s -> FourierPoints[1, i1, p, q]).
          (Transpose[A] /. s → FourierPoints[1, i1, p, q])], 10^(-10)];
       ,
       DrazinInverse,
   res = Chop[Det[FourierPoints[2, i2, p, q] * IdentityMatrix[p] -
         (A /. s -> FourierPoints[1, i1, p, q])], 10^(-10)];
  ];
  Chop[res]
 ]

CalculateDeterminant[A_, opts___?OptionQ] :=
 Module[{n1, n2, p, m, q, in, tbl, deflt, method},
  deflt = {Method → GeneralizedInverse};
  {method} = {Method} /. Flatten[{opts}] /. deflt;
  {p, m} = Dimensions[A];
  q = Max[Append[Flatten[Exponent[A, s]], 0]];
  n1 = 2 * p * q;
  n2 = p;
  tbl = Table[Chop[CalculateDeterminantCoefficient[A, i - 1,
      j - 1, p, m, q, Method -> method], 10^(-10)], {i, n1 + 1}, {j, n2 + 1}];
  in = InverseFourier[tbl, FourierParameters -> {1, -1}] // N;
  in = Chop[in, 10^(-10)] // N;
  RRound[in, 10^(-6)]
 ]

CalculateBCoefficient[A_, i_Integer, p_Integer, q_Integer, ais_, l_Integer] :=
 Module[{lc, o, j, AT, si, Asi, ATsi, aissi, B, ss, no},
  AT = Transpose[A];
  si = FourierPoints[3, i, p, q];
  Asi = A /. s → si;
  ATsi = AT /. s → si;
  ss = Table[Chop[N[si^(o - 1)], 10^(-10)], {o, Dimensions[ais][[1]]}];
  aissi = Chop[ss.ais, 10^(-10)];
  no = Dimensions[ais][[2]];
  B = Chop[ATsi.(((Sum[(aissi[[no - j - 1]]) * MatrixPower[((Asi).(ATsi)), (l - 2 - j)], {j, 0, l-2}]) +
       MatrixPower[((Asi).ATsi), (l - 1)]), 10^(-10)]
```

```
 ]

GeneralizedInverse[A_] :=
 Module[{lc, B, p, m, q, n, dt, ais, ss, k},
  {p, m} = Dimensions[A];
  q = Max[Exponent[A, s]];
  n = (2 * p - 1) * q;
  ais = CalculateDeterminant[A];
  k = CalculateK[ais] - 1;
  If[k > 0,
   lc = Table[CalculateBCoefficient[A, i - 1, p, q, ais, k], {i, n + 1}];
   ss = (Table[s^(o - 1), {o, Dimensions[ais][[1]]}]).
     ais[[All, {Dimensions[ais][[2]] - k}]];
   Chop[-CalculateInverseFourierTransform[lc, m, p] / ss[[1]], 10^(-10)],
   ss = Table[0, {o, m}, {i, p}]]
 ]

MakePolynomial[ls_] :=
 Module[{},
  Plus @@ MapIndexed[Function[{x, y}, x * s^(y[[1]] - 1)], ls, 1]
 ]

CalculateInverseFourierTransform[a_, p_Integer, m_Integer] :=
 Module[{quo, i, r, j},
  quo = Quotient[Times @@ Dimensions[a], p * m];
  Table[MakePolynomial[RRound[Chop[InverseFourier[
       Flatten[a][[Table[(r - 1) * (m) + j + p * m * (i - 1), {i, quo}]] // Flatten]],
       FourierParameters -> {1, -1}], 10^(-10)], 10^(-5)]], {r, p}, {j, m}]
 ]

CalculateK[a_] :=
 Module[{m, lst},
  m = Dimensions[a][[2]];
  lst = Table[And @@ (LessThane[a[[All, {i}]]] // Flatten), {i, m}];
  -Min[Append[Position[lst, False], m + 1] // Flatten] + 1 + m
 ]

LessThane[a_] :=
 Module[{},
  If[Abs[a] <= 10^(-10), True, False]
 ]
SetAttributes[LessThane, Listable]

RRound[a_, b_: 10^(-10)] :=
 Module[{ra},
  ra = Round[a];
  If[Abs[ra] - Abs[a] < b, ra, a]
 ]
SetAttributes[RRound, Listable]

CheckIfIsZero[a_] :=
 Module[{q, i, b},
  (* For simplicity we do not use Fourier Points but integers *)
  q = Max[Exponent[A, s]];
  b = Table[a /. s -> i, {i, q + 1}];
```

```
  And @@ Flatten[LessThane[b]]
 ]


DrazinFourierPoints[ri_Integer, nj_Integer] :=
 Module[{M, lc},
  lc = Chop[N[Exp[(-2 * ri * Pi * I) / (nj + 1)]], 10^(-10)]
 ]

CalculateDrazinK[a_] :=
 Module[{m, lst},
  m = Dimensions[a][[2]];
  lst = Table[And @@ (LessThane[a[[All, {i}]]] // Flatten), {i, m}];
  -Min[Append[Position[lst, False], m + 1] // Flatten] + 1 + m
 ]

CalculateDrazinBCoefficient[A_, i_Integer,
  q_Integer, ais_, l_Integer, nj_Integer, k_: 0] :=
 Module[{lc, o, j, AT, si, Asi, aissi, B, ss, no},
  si = DrazinFourierPoints[i, nj];
  Asi = A /. s → si;
  ss = Table[Chop[N[si^(o - 1)], 10^(-10)], {o, Dimensions[ais][[1]]}];
  aissi = Chop[ss.ais, 10^(-10)];
  no = Dimensions[ais][[2]];
  B = Chop[MatrixPower[Asi, k].
      (MatrixPower[(((Sum[(aissi[[no - j]]) * MatrixPower[(Asi), (l - j)], {j, 1, l}]) +
          MatrixPower[(Asi), (l)]), k + 1])]
 ]

DrazinInverse[A_] :=
 Module[{dt, t, m, lc, Bj, q, r, k, j, i, B, Bjf, ats, test, nj},
  m = Dimensions[A][[1]];
  q = Max[Exponent[A, s]];
  If[m == Dimensions[A][[2]],
   dt = CalculateDeterminant[A, Method → DrazinInverse];
   t = CalculateK[dt] - 1;
   j = m;
   nj = 2 * j * q;
   Bjf = Table[CalculateDrazinBCoefficient[A, i - 1, q, dt, j, nj], {i, nj + 1}];
   Bj = CalculateInverseFourierTransform[Bjf, m, m];
   test = CheckIfIsZero[Bj];
   While[test,
    j = j - 1;
    Bjf = Table[CalculateDrazinBCoefficient[A, i - 1, q, dt, j, nj], {i, nj + 1}];
    Bj = CalculateInverseFourierTransform[Bjf, m, m];
    test = CheckIfIsZero[Bj];
   ];
   r = j + 1;
   k = r - t;
   nj = 2 * (t - 1) * q * (k + 1) + q * k;
   B = Table[CalculateDrazinBCoefficient[A, i - 1, q, dt, t - 1, nj, k], {i, nj + 1}];
```

```
  B = CalculateInverseFourierTransform[B, m, m];
  ats = (Table[s^(o - 1), {o, Dimensions[dt][[1]]}]).
    dt[[All, {Dimensions[dt][[2]] - t}]];
  B / ((-ats[[1]])^(k + 1))
  ,
  Print["Argument must be a square matrix"]
 ]
]
```